
PROGRAMA E BLOCO

<programa> ::=
program <identificador> ; <bloco> .

<bloco> ::=
<parte de declarações de variáveis> <comando composto>

DECLARAÇÕES

<parte de declarações de variáveis> ::=
{ **var** <declaração de variáveis> { ; <declaração de variáveis> } ; }

<declaração de variáveis> ::=
<lista de identificadores> : <tipo>

<lista de identificadores> ::=
<identificador> { , <identificador> }

<tipo> ::=
integer | **real**

COMANDOS

<comando composto> ::=
begin <comando> { ; <comando> } **end**

<comando> ::=
<atribuição> | <comando composto> | <comando condicional> | <comando repetitivo>

<atribuição> ::=
<variável> := <expressão>

<comando condicional> ::=
if <expressão> **then** <comando> [**else** <comando>]

<comando repetitivo> ::=
while <expressão> **do** <comando>

EXPRESSÕES

<expressão> ::=
<expressão simples> [<relação> <expressão simples>]

<relação> ::=
= | <> | < | <= | >= | >

<expressão simples> ::=
[+ | -] <termo> { (+ | -) <termo> }

<termo> ::=
<fator> { (* | /) <fator> }

<fator> ::=
<variável> | <número> | (<expressão>)

<variável> ::=
<identificador>

IDENTIFICADORES E NÚMEROS

$\langle \text{identificador} \rangle ::= \langle \text{letra} \rangle (\langle \text{letra} \rangle \mid \langle \text{dígito} \rangle)^*$

$\langle \text{letra} \rangle ::= a \mid b \mid c \mid \dots \mid z \mid A \mid B \mid C \mid \dots \mid Z$

$\langle \text{dígito} \rangle ::= 0 \mid 1 \mid 2 \mid 3 \mid 4 \mid 5 \mid 6 \mid 7 \mid 8 \mid 9$

$\langle \text{inteiro sem sinal} \rangle ::= \langle \text{dígito} \rangle (\langle \text{dígito} \rangle)^*$

$\langle \text{sinal} \rangle ::= + \mid - \mid \varepsilon$

$\langle \text{fator de escala} \rangle ::= \text{“E”} \langle \text{sinal} \rangle \langle \text{inteiro sem sinal} \rangle$

$\langle \text{real sem sinal} \rangle ::= \langle \text{inteiro sem sinal} \rangle (\varepsilon \mid \text{“.”} \langle \text{dígito} \rangle^*) (\varepsilon \mid \langle \text{fator de escala} \rangle^*)$

$\langle \text{número} \rangle ::= \langle \text{sinal} \rangle \langle \text{inteiro sem sinal} \rangle \mid \langle \text{sinal} \rangle \langle \text{real sem sinal} \rangle$